Using Spice to Simulate Gain Dynamics in Doped-fiber Amplifier Chains

Presented at OFC'98 Workshop 204, Monday Feb. 23, 1998

Alberto Bononi (alberto@tlc.unipr.it), Lorenzo Barbieri
Università di Parma, Dipartimento di Ingegneria dell'Informazione, Parma, I-43100, Italy.
Leslie A. Rusch
Université Laval, Dépt. de génie électrique et informatique, Ste-Foy, Québec, G1K 7P4, Canada.

Abstract— *It has recently been shown that gain dynamics in doped-fiber amplifiers can be simulated by an equivalent electric circuit [1]. Here we include ASE-induced self-saturation in the model, and discuss its implementation using SPICE.*

I. Theory

Building on the results of [2], and under the same assumptions of a two-level system for the dopant ions, it has recently been shown that gain dynamics in doped-fiber amplifiers can be completely described by the total number of excited ions in the amplifier, which we call the *reservoir* r [1]. Here we extend the results in [1] by including self-saturation by spontaneous emission (ASE).

The time evolution of r is described by the following first-order nonlinear differential equation involving the input signal photon fluxes $Q_j(t)$ [photons/s] at frequencies ν_j, $j = 0,..,N$, of an N channel WDM system (channel 0 representing the pump):

$$\dot{r}(t) = \sum_{j=0}^{N} Q_j(t) - \left[\sum_{j=0}^{N} Q_j(t) g_j(r(t)) + \frac{r(t)}{\tau} + Q_{ASE}(r(t)) \right]. \tag{1}$$

Such equation, derived in [1] without ASE, states that the time variation of the reservoir r equals the input photon fluxes, minus the output photon fluxes, namely, amplified signal fluxes, spontaneous emission, and ASE. The gain at frequency ν_j is [1]: $g_j(r) = e^{B(\nu_j)\, r - A(\nu_j)}$, where $B(\nu) \triangleq \Gamma(\nu)\sigma^T(\nu)/A_{eff}$ and $A(\nu) \triangleq \rho L \Gamma(\nu)\sigma^a(\nu)$ and are non-dimensional parameters; ρ [m^{-3}] is the ion density in the doped fiber core of effective area A_{eff} [m^2]; $\Gamma(\nu)$, $\sigma^e(\nu)$ [m^2], and $\sigma^a(\nu)$ [m^2] are the confinement factor, emission and absorption cross-sections at frequency ν, respectively, and $\sigma^T(\nu) \triangleq \sigma^e(\nu) + \sigma^a(\nu)$; L [m] is the length of the amplifier, and τ [s] is the fluorescence time.

Assuming a constant inversion $x \triangleq r/r_M$, where $r_M = \rho A_{eff} L$ is the total number of ions, we can express the spontaneous emission factor as [3]

$$n_{sp}(r,\, \nu_j) = \frac{\sigma^e(\nu_j)\, x}{\sigma^T(\nu_j)\, x - \sigma^a(\nu_j)} = \frac{B(\nu_j) - A(\nu_j)/r_M}{\ln\left(g_j(r)\right)}\, r \tag{2}$$

and thus approximate the ASE as [4, eq. (5)]:

$$Q_{ASE}(r) = \sum_{m=1}^{M} 4(g_m(r) - 1)\, n_{sp}(r,\, \nu_m)\Delta\nu_m \tag{3}$$

where the factor 4 takes into account forward and backward ASE, with two polarization components each, and the summation is calculated over the frequency bands $\Delta\nu_m$, $m = 1,..,M$ where ASE takes non-negligible values.

II. Circuit equivalent in SPICE

By interpreting r as the charge on a capacitor C of voltage v: $r \triangleq Cv$ and the photon fluxes as currents, we have an



Fig. 1. Doped-fiber amplifier equivalent circuit with ASE.

immediate interpretation of eq. (1) as a Kirkhoff current law at a node [1]. To fit the parameter values within SPICE's allowed range it may be necessary to divide both sides of (1) by a scaling factor F. Thus defining:

$$v'(t) \triangleq \frac{v(t)}{F} = \frac{r(t)}{CF}; \quad i'_j(t) \triangleq \frac{Q_j(t)}{F}; \quad B'(\nu) \triangleq B(\nu)CF \tag{4}$$

we obtain the Kirkhoff current law equation:

$$C\dot{v'} = -\frac{v'}{R} + \sum_{j=0}^{N} i'_j(1 - g_j(v')) - \sum_{m=1}^{M} i'_{\mathrm{ASE}m} \tag{5}$$

where $R = \tau/C$, $g_j(v') \triangleq e^{B'(\nu_j)v' - A(\nu_j)}$, and the ASE terms are $i'_{\mathrm{ASE}m} \triangleq \frac{4}{F}(g_m(v') - 1)\frac{(B'(\nu_m) - \frac{A(\nu_m)}{y_M})}{\ln\left(g_m(v')\right)}\, v'\, \Delta\nu_m$, where $y_M \triangleq r_M/(FC)$.

Equation (5) is implemented by the electric circuit shown in Fig. 1, where we also considered the input ASE fluxes coming from previous amplifiers. ASE and SIGNAL fluxes can be treated separately, even when they are on the same frequency band. This allows monitoring the signal to noise ratio (SNR) along a chain of amplifiers. In the circuit, the natural time constant in the absence of signals is the fluorescence time τ. Each amplified output signal current is a nonlinear current generator driven by the normalized reservoir v' and the corresponding input current.

III. Example

We simulated in SPICE an example of channel drop in a chain of 5 EDFAs and 2 input signals presented in [5], using similar parameters. The amplifiers have two input channels $\lambda_1 = 1552.4$ nm and $\lambda_2 = 1557.9$ nm, with initial input powers -3 dBm. Fig. 2 shows the M=6 ASE bands and the N=2 WDM signals used in SPICE. Fig. 3 shows the SPICE code for the EDFA subcircuit. The system is at equilibrium before $t = 0$. At time $t = 0$ channel 2 is dropped.

Fig. 5 gives the output current corresponding to the surviving channel. The photon flux is obtained by multiplying such current by the scaling factor 10^{16}. The results match



Fig. 2. Wavelengths bands used in the example.

```
* EDFA equivalent electrical circuit, 2 channels, 6 ASE noise bands
* in nodes: 1 2 3 lower out-of-band ASE noise, 4 5 channel 1 and in-band
*           ASE noise, 6 7 channel 2 and in-band ASE noise, 8 upper out-
*           of-band ASE noise; 9 reservoir (internal)
* out nodes: 10 11 12 lower out-of band ASE noise, 13 14 channel 1 and
*           in-band ASE noise, 15 16 channel 2 and in-band ASE noise,
*           17 upper out-of-band ASE noise

* EDFA, Pump and channels parameters:

.PARAM FACT = 1E16   ;reservoir & flux scaling factor
.PARAM CAP = 1E-2    ;capacitance
.PARAM tau = 10.5E-3 ;fluorescence time
.PARAM Iinp = 34.112 ;=Qp/FACT, Pp=18.4dBm, 980nm, Qp=34.112E16 pump flux
.PARAM Ap  = 8.995
.PARAM Bsp = 4.390 ;=Bp*CAP*FACT Bp=4.390E-14
.PARAM A1  = 5.075
.PARAM Bs1 = 6.190 ;=B1*CAP*FACT B1=6.190E-14
.PARAM A2  = 4.375
.PARAM Bs2 = 5.678 ;=B2*CAP*FACT B2=5.678E-14

* ASE noise parameters (Bai=Bi*CAP*FACT):

.PARAM Aa1 = 5.8685
.PARAM Ba1 = 4.7286
.PARAM Aa2 = 10.5326
.PARAM Ba2 = 10.7873
.PARAM Aa3 = 8.8854
.PARAM Ba3 = 8.5420
.PARAM Aa4 = 5.1149
.PARAM Ba4 = 6.1452
.PARAM Aa5 = 4.1745
.PARAM Ba5 = 5.6529
.PARAM Aa6 = 0.5617
.PARAM Ba6 = 1.1778

.PARAM Dni = 6.2435E11 ;delta-ni=5nm @ 1550nm (considered as constant)
.PARAM yM  = 2.049 ;=rM/CAP/FACT   rM=2.049E14

.FUNC Iasel1(y) 4*(4*Dni)*y*(EXP(Ba1*y-Aa1)-1)*(Ba1-Aa1/yM)/(Ba1*y-Aa1)/FACT
.FUNC Iasel2(y) 4*(3*Dni)*y*(EXP(Ba2*y-Aa2)-1)*(Ba2-Aa2/yM)/(Ba2*y-Aa2)/FACT
.FUNC Iasel3(y) 4*(2*Dni)*y*(EXP(Ba3*y-Aa3)-1)*(Ba3-Aa3/yM)/(Ba3*y-Aa3)/FACT
.FUNC Iase1(y) 4*(2*Dni)*y*(EXP(Ba4*y-Aa4)-1)*(Ba4-Aa4/yM)/(Ba4*y-Aa4)/FACT
.FUNC Iase2(y) 4*(2*Dni)*y*(EXP(Ba5*y-Aa5)-1)*(Ba5-Aa5/yM)/(Ba5*y-Aa5)/FACT
.FUNC Iaseu1(y) 4*(15*Dni)*y*(EXP(Ba6*y-Aa6)-1)*(Ba6-Aa6/yM)/(Ba6*y-Aa6)/FACT

.SUBCKT EDFA 1 2 3 4 5 6 7 8   10 11 12 13 14 15 16 17

Ip    0  9 {Iinp}
R1     1  9 1OHM   ; resistors to convert current to voltage
R2     2  9 1OHM
R3     3  9 1OHM
R4     4  9 1OHM
R5     5  9 1OHM
R6     6  9 1OHM
R7     7  9 1OHM
R8     8  9 1OHM
Ry     9  0 {tau/CAP}
Cy     9  0 {CAP}
G0     9  0 VALUE = { Iinp*EXP(Bsp*V(9)-Ap) }
Gasel1 9 10 VALUE = { V(1,9)*EXP(Ba1*V(9)-Aa1)+Iasel1(V(9)) }
Gasel2 9 11 VALUE = { V(2,9)*EXP(Ba2*V(9)-Aa2)+Iasel2(V(9)) }
Gasel3 9 12 VALUE = { V(3,9)*EXP(Ba3*V(9)-Aa3)+Iasel3(V(9)) }
G1     9 13 VALUE = { V(4,9)*EXP(Bs1*V(9)-A1) }
Gase1  9 14 VALUE = { V(5,9)*EXP(Ba4*V(9)-Aa4)+Iase1(V(9)) }
G2     9 15 VALUE = { V(6,9)*EXP(Bs2*V(9)-A2) }
Gase2  9 16 VALUE = { V(7,9)*EXP(Ba5*V(9)-Aa5)+Iase2(V(9)) }
Gaseu1 9 17 VALUE = { V(8,9)*EXP(Ba6*V(9)-Aa6)+Iaseu1(V(9)) }

.ENDS EDFA
```

Fig. 3. SPICE code for the EDFA subcircuit.

```
Cascaded EDFAs(5), with ASE - 2 channels

.PARAM Iin1 = 1.5584 ;P1=3dBm, 1552.4nm, Q1=1.5584E16
.PARAM Iin2 = 1.5639 ;P2=3dBm, 1557.9nm, Q2=1.5639E16

Iasel1  0  1 0A
Iasel2  0  2 0A
Iasel3  0  3 0A
I1      0  4 {Iin1}
Iase1   0  5 0A
I2      0  6 PULSE({Iin2} 0A 0S 0S 0S 0.5MS 1MS)
Iase2   0  7 0A
Iaseu1  0  8 0A

X1    1  2  3  4  5  6  7  8    9 10 11 12 13 14 15 16 EDFA
XL1   9 10 11 12 13 14 15 16   17 18 19 20 21 22 23 24 LOSS
X2   17 18 19 20 21 22 23 24   25 26 27 28 29 30 31 32 EDFA
XL2  25 26 27 28 29 30 31 32   33 34 35 36 37 38 39 40 LOSS
X3   33 34 35 36 37 38 39 40   41 42 43 44 45 46 47 48 EDFA
XL3  41 42 43 44 45 46 47 48   49 50 51 52 53 54 55 56 LOSS
X4   49 50 51 52 53 54 55 56   57 58 59 60 61 62 63 64 EDFA
XL4  57 58 59 60 61 62 63 64   65 66 67 68 69 70 71 72 LOSS
X5   65 66 67 68 69 70 71 72   73 74 75 76 77 78 79 80 EDFA

RO1  73 0 1OHM ; output resistance to measure output current
RO2  74 0 1OHM
RO3  75 0 1OHM
RO4  76 0 1OHM
RO5  77 0 1OHM
RO6  78 0 1OHM
RO7  79 0 1OHM
RO8  80 0 1OHM

* Set initial conditions
.NODESET V(X1.9)=1.2 V(X2.9)=1.2 V(X3.9)=1.2
.TRAN 1US 350US
.PROBE
.END
```

Fig. 4. SPICE main code for the EDFA chain simulation.



Fig. 5. Surviving channel output current at amplifier 1 through 5; multiplied by 10^{16} it gives the photon flux.



Fig. 6. SNR (dB) time evolution on the surviving channel at amplifier 1 through 5, measured over an optical bandwidth of 1 nm.

with those obtained by direct numerical integration. Fig. 6 finally shows the time evolution of the optical SNR (the ASE power is over a bandwidth of 1 nm) at the surviving channel wavelength. It is interesting to observe that, although the output power has oscillations at the output of amplifiers 2–5, even the ASE has, so that the SNR has a smooth time evolution.

References

[1] A. Bononi, L. A. Rusch, and L. Tancevski, "A simple dynamic model of fiber amplifiers and its equivalent electrical circuit," submitted to *Electron. Lett.*, Aug. 1997.

[2] Y. Sun, G. Luo, J. L. Zyskind, A. A. M. Saleh, A. K. Srivastava, and J. W. Sulhoff, "Model for gain dynamics in erbium-doped fibre amplifiers," *Electron. Lett.*, vol. 32, pp. 1490–1491, Aug. 1996.

[3] E. Desurvire, *Erbium-doped fiber amplifiers.* Wiley, 1994.

[4] T. Georges, and E. Delevaque, "Analytic modeling of high-gain erbium-doped fiber amplifiers," *Opt. Lett.*, vol. 17, pp. 1113–1115, Aug. 1992.

[5] Y. Sun, A. K. Srivastava, J. L. Zyskind, J. W. Sulhoff, J. Wolf, and R. W. Tkach, "Fast power transients in WDM optical networks with cascaded EDFAs," *Electron. Lett.*, vol. 33, pp. 313–314, Feb. 1997.